MLC Lab Visit - Lab 06 - Maple

Mth 355 (a.k.a. Mth 399) Feb 12, 2003 Maple 7

Bent E. Petersen

petersen@math.orst.edu

There are 3 problems below. Problem solutions are due Feb 19, 2003. Email your solutions to me as Maple worksheet attachments. Your worksheet must execute correctly for full credit.

In this week's lab we investigate a few randomly chosen features of Maple.

Partial Fractions

```
> restart;
```

Partial fraction decomposition in Maple is carried out with the convert() command.

```
> f:=(3*x^2-2*x+7)/( (x^2+1)*(x-3)^3*(x^2+x-1) );
```

$$f := \frac{3x^2 - 2x + 7}{(x^2+1)(x-3)^3(x^2+x-1)}$$

```
> convert(f,parfrac,x);
```

$$\frac{14}{55}\frac{1}{(x-3)^3} - \frac{512}{3025}\frac{1}{(x-3)^2} + \frac{\frac{30197}{332750}}{x-3} + \frac{\frac{1}{250}(9+13x)}{x^2+1} - \frac{5}{1331}\frac{75+38x}{x^2+x-1}$$

Note since all the coefficients here are rational, Maple attempts the expansion over the rational numbers. This is the reason the last term is not what you were led to expect in calculus since x^2+x-1 does factor over the rationals. We can force a factorization by telling Maple to use the extension field of the rationals obtained by appending the square root of 5.

```
> convert(f,parfrac,x,sqrt(5));
```

$$\frac{2}{1331}\frac{56\sqrt{5}-95}{2x+1+\sqrt{5}} - \frac{2}{1331}\frac{56\sqrt{5}+95}{2x+1-\sqrt{5}} + \frac{\frac{1}{250}(9+13x)}{x^2+1} + \frac{\frac{30197}{332750}}{x-3} - \frac{512}{3025}\frac{1}{(x-3)^2} + \frac{\frac{14}{55}}{(x-3)^3}$$

This expression should agree with what you learned in calculus, but it is inconvenient, and not really general, to have to specifiy the radicals to append. A more general and more convenient procedure is to use RootOf::

```
> h:=convert(f,parfrac,x,RootOf(x^2+x-1));
```

$$h := -\frac{1}{1331}\,\frac{151+112\,\mathrm{RootOf}(\_Z^2+\_Z-1)}{x-\mathrm{RootOf}(\_Z^2+\_Z-1)}+\frac{\frac{1}{1331}\,(-39+112\,\mathrm{RootOf}(\_Z^2+\_Z-1))}{x+1+\mathrm{RootOf}(\_Z^2+\_Z-1)}+\frac{\frac{1}{250}\,(9+13\,x)}{x^2+1}+\frac{\frac{30197}{332750}}{x-3}-\frac{512}{3025}\,\frac{1}{(x-3)^2}+\frac{\frac{14}{55}}{(x-3)^3}$$

Cool, but not very familiar. We need to evaluate the RootOf() expressions. This may be done with the allvalues() command, but it will evaluate the full expression for each root (and so give us 2 copies):

```
> allvalues(h);
```

$$-\frac{1}{1331}\,\frac{56\sqrt{5}+95}{x+\frac{1}{2}-\frac{1}{2}\sqrt{5}}+\frac{\frac{1}{1331}\,(56\sqrt{5}-95)}{x+\frac{1}{2}+\frac{1}{2}\sqrt{5}}+\frac{\frac{1}{250}\,(9+13\,x)}{x^2+1}+\frac{\frac{30197}{332750}}{x-3}-\frac{512}{3025}\,\frac{1}{(x-3)^2}+\frac{\frac{14}{55}}{(x-3)^3},\ -\frac{1}{1331}\,\frac{95-56\sqrt{5}}{x+\frac{1}{2}+\frac{1}{2}\sqrt{5}}+\frac{\frac{1}{1331}\,(-95-56\sqrt{5})}{x+\frac{1}{2}-\frac{1}{2}\sqrt{5}}+\frac{\frac{1}{250}\,(9+13\,x)}{x^2+1}+\frac{\frac{30197}{332750}}{x-3}-\frac{512}{3025}\,\frac{1}{(x-3)^2}+\frac{\frac{14}{55}}{(x-3)^3}$$

As expected we got 2 copies of the answer. Just pick one:

```
> allvalues(h)[1];
```

$$-\frac{1}{1331}\,\frac{56\sqrt{5}+95}{x+\frac{1}{2}-\frac{1}{2}\sqrt{5}}+\frac{\frac{1}{1331}\,(56\sqrt{5}-95)}{x+\frac{1}{2}+\frac{1}{2}\sqrt{5}}+\frac{\frac{1}{250}\,(9+13\,x)}{x^2+1}+\frac{\frac{30197}{332750}}{x-3}-\frac{512}{3025}\,\frac{1}{(x-3)^2}$$

$$+\frac{\frac{14}{55}}{(x-3)^3}$$

We can also request that Maple perform the factorization over the real numbers. In this case Maple will use floating point arithmetic.

```
> convert(f,parfrac,x,real);
```

$$.02270458805\,\frac{1}{x+1.618033989}-\frac{.1654544002}{x-.6180339887}+\frac{.2545454546}{(x-3.)^3}-\frac{.1692561984}{(x-3.)^2}$$
$$+\frac{.09074981217}{x-3.}+\frac{.03599999996+.05200000000\,x}{x^2+1.}$$

You can make a rational approximation in the obvious way:

```
> convert(convert(f,parfrac,x,real),rational);
```

$$\frac{6119}{269505}\,\frac{1}{x+\frac{28657}{17711}}-\frac{2070}{12511}\,\frac{1}{x-\frac{17711}{28657}}+\frac{\frac{14}{55}}{(x-3)^3}-\frac{512}{3025}\,\frac{1}{(x-3)^2}+\frac{\frac{27659}{304783}}{x-3}$$
$$+\frac{\frac{3599996}{99999889}+\frac{13}{250}\,x}{x^2+1}$$

Let's look at complex roots:

```
> g:=( (2*x^3-x-1)/((x^2+2*x+5)^2*(x^2-1)) );
```

$$g:=\frac{2\,x^3-x-1}{(x^2+2\,x+5)^2\,(x^2-1)}$$

```
> convert(g,parfrac,x);
```

$$\frac{1}{16}\,\frac{1}{x+1}-\frac{1}{16}\,\frac{x+1}{x^2+2\,x+5}+\frac{\frac{1}{4}(-1+7\,x)}{(x^2+2\,x+5)^2}$$

Exactly what we expected! Of course we may prefer to expand over the complex numbers:

```
> convert(g,parfrac,x,complex);
```

$$\frac{.1250000000+.2187500000\,I}{(x+1.000000000+2.000000000\,I)^2}-\frac{.03125000000+.06250000000\,I}{x+1.+2.\,I}+\frac{.06250000000}{x+1.}+\frac{.1250000000-.2187500000\,I}{(x+1.-2.000000000\,I)^2}-\frac{.03125000000-.06250000000\,I}{x+1.-2.\,I}$$

Hmm. We got a floating point expansion. We can convert to rational numbers in this case:

```
> convert(convert(g,parfrac,x,complex),rational);
```

$$\frac{\frac{1}{8}+\frac{7}{32}I}{(x+1+2I)^2}-\frac{\frac{1}{32}+\frac{1}{16}I}{x+1+2I}+\frac{\frac{1}{16}}{x+1}+\frac{\frac{1}{8}-\frac{7}{32}I}{(x+1-2I)^2}-\frac{\frac{1}{32}-\frac{1}{16}I}{x+1-2I}$$

That looks alright here, but sometimes the conversion from floating point to rational numbers will introduce errors in the final expression. It's safer to make use of RootsOf().

```
> allvalues(convert(g,parfrac,x,RootOf(x^2+2*x+5)))[1];
```

$$\frac{\frac{1}{8}+\frac{7}{32}I}{(x+1+2I)^2}-\frac{\frac{1}{32}+\frac{1}{16}I}{x+1+2I}+\frac{\frac{1}{16}}{x+1}+\frac{\frac{1}{8}-\frac{7}{32}I}{(x+1-2I)^2}-\frac{\frac{1}{32}-\frac{1}{16}I}{x+1-2I}$$

Another way, perhaps the best way, to handle rational complex roots is to specify our extension field as 'sqrt(-1)' rather than 'complex'.

```
> convert(g,parfrac,x,sqrt(-1));
```

$$\frac{\frac{1}{8}+\frac{7}{32}I}{(x+1+2I)^2}-\frac{\frac{1}{32}+\frac{1}{16}I}{x+1+2I}+\frac{\frac{1}{16}}{x+1}+\frac{\frac{1}{8}-\frac{7}{32}I}{(x+1-2I)^2}-\frac{\frac{1}{32}-\frac{1}{16}I}{x+1-2I}$$

A more ambitious example

The partial fraction expansion facility is not limited to simple examples, but one pretty well has to use floating point, or a mixture of floating point and rational arithmetic, except in special cases.

```
> p:=x^9+5*x^8+10*x^7+20*x^6+24*x^5-x^4+16;
```

$$p := x^9+5\,x^8+10\,x^7+20\,x^6+24\,x^5-x^4+16$$

```
> convert(1/p,parfrac,x);
```

$$-\frac{1}{224}\frac{1}{x+2}-\frac{1}{12680}\frac{39+4\,x}{x^2+4}$$

$$+\frac{\frac{1}{355040}(46511-23992\,x+628\,x^2-342\,x^3+3013\,x^4+1697\,x^5)}{x^6+3\,x^5-x+2}$$

From our expeience above we know the problem here is that $x^6+3\,x^5-x+2$ is irreducible over the rationals. We can try RootOf($x^6+3\,x^5-x+2$) as our extension field, but we only obtain a result which is not very satisfactory as a final result, though it may be useful for further calculation. The best thing to do probably, is to use floating point, real or complex according to our needs or taste.

```
> cr:=convert(1/p,parfrac,x,real);
```

$$cr := .0003493765388\,\frac{1}{x+2.978770645}-\frac{.004464285719}{x+2.000000000}+\frac{.01871132547}{x+1.103523276}+\frac{.02416405767+.001706084210\,x}{x^2+.3190908316\,x+.9087228666}-\frac{.003075709712+.0003154571462\,x}{x^2+4.}-\frac{-.01462057076+.01598704311\,x}{x^2-1.401384753\,x+.6695453152}$$

```
> cf:=convert(1/p,parfrac,x,complex);
```

$$cf := \frac{.0003493765390-.1240884475\,10^{-13}\,I}{x+2.978770645}-\frac{.004464285715+.7441024610\,10^{-13}\,I}{x+2.}+\frac{.01871132547-.3975527969\,10^{-11}\,I}{x+1.103523276}+\frac{.0008530421083+.01271082328\,I}{x+.1595454158+.9398234552\,I}+\frac{.0008530421122-.01271082329\,I}{x+.1595454158-.9398234552\,I}-\frac{.0001577287066+.0007689274447\,I}{x+2.\,I}-\frac{.0001577287068-.0007689274446\,I}{x-2.\,I}-\frac{.007993521556-.004044862196\,I}{x-.7006923766+.4225819549\,I}-\frac{.007993521552+.004044862196\,I}{x-.7006923766-.4225819549\,I}$$

The very small imaginary parts are due to roundoff (the corresponding terms should be real). Let's eliminate these small imaginary parts:

```
> subs(seq(op(cf)[k]=evalc(Re(op(cf)[k])),k=1..3),cf);
```

$$.0003493765390\,\frac{1}{x+2.978770645}-\frac{.004464285715}{x+2.}+\frac{.01871132547}{x+1.103523276}+\frac{.0008530421083+.01271082328\,I}{x+.1595454158+.9398234552\,I}+\frac{.0008530421122-.01271082329\,I}{x+.1595454158-.9398234552\,I}$$

$$-\frac{.0001577287066+.0007689274447\,I}{x+2.\,I}-\frac{.0001577287068-.0007689274446\,I}{x-2.\,I}$$
$$-\frac{.007993521556-.004044862196\,I}{x-.7006923766+.4225819549\,I}-\frac{.007993521552+.004044862196\,I}{x-.7006923766-.4225819549\,I}$$

Another approach is to split off the terms we can evaluate exactly:

```
> c2f:=convert(1/p,parfrac,x,sqrt(-1));
```

$$c2f:=\frac{\frac{1}{6340}-\frac{39}{50720}I}{-x+2\,I}-\frac{\frac{1}{6340}+\frac{39}{50720}I}{x+2\,I}-\frac{1}{224}\frac{1}{x+2}+\frac{\frac{1}{355040}(46511-23992\,x+628\,x^2-342\,x^3+3013\,x^4+1697\,x^5)}{x^6+3\,x^5-x+2}$$

Note using 'sqrt(-1)' rather than 'complex' forces Maple to use rational arithmetic rather than floating point.

```
> c2p:=op(c2f)[4];
```

$$c2p:=\frac{1}{355040}\frac{46511-23992\,x+628\,x^2-342\,x^3+3013\,x^4+1697\,x^5}{x^6+3\,x^5-x+2}$$

```
> c3f:=convert(c2p,parfrac,x,complex);
```

$$c3f:=\frac{.0003493765384+.2175991760\,10^{-14}\,I}{x+2.978770645}+\frac{.01871132547+.9872910312\,10^{-12}\,I}{x+1.103523276}$$
$$+\frac{.0008530421057+.01271082329\,I}{x+.1595454158+.9398234552\,I}-\frac{.0008530421046-.01271082329\,I}{-1.\,x-.1595454158+.9398234552\,I}$$
$$-\frac{.007993521556-.004044862199\,I}{x-.7006923766+.4225819549\,I}+\frac{.007993521556+.004044862199\,I}{-1.\,x+.7006923766+.4225819549\,I}$$

```
> c4f:=subs(seq(op(c3f)[k]=evalc(Re(op(c3f)[k])),k=1..2),c3f);
```

$$c4f:=.0003493765384\frac{1}{x+2.978770645}+\frac{.01871132547}{x+1.103523276}$$
$$+\frac{.0008530421057+.01271082329\,I}{x+.1595454158+.9398234552\,I}-\frac{.0008530421046-.01271082329\,I}{-1.\,x-.1595454158+.9398234552\,I}$$
$$-\frac{.007993521556-.004044862199\,I}{x-.7006923766+.4225819549\,I}+\frac{.007993521556+.004044862199\,I}{-1.\,x+.7006923766+.4225819549\,I}$$

```
>
> sum(op(c2f)[k],k=1..3)+evalf(c4f,6);
```

$$\frac{\frac{1}{6340}-\frac{39}{50720}I}{-x+2I}-\frac{\frac{1}{6340}+\frac{39}{50720}I}{x+2I}-\frac{1}{224}\frac{1}{x+2}+\frac{.000349377}{x+2.97877}+\frac{.0187113}{x+1.10352}$$
$$+\frac{.000853042+.0127108\,I}{x+.159545+.939823\,I}-\frac{.000853042-.0127108\,I}{-1.\,x-.159545+.939823\,I}-\frac{.00799352-.00404486\,I}{x-.700692+.422582\,I}$$
$$+\frac{.00799352+.00404486\,I}{-1.\,x+.700692+.422582\,I}$$

Not exact perhaps, but it is a useful result.

```
>
```

Full Partial Fractions

```
> restart;
```

The fullparfrac facility performs a partial fraction decomposition of a rational expression in one variable after completely factoring the denominator into linear factors over its splitting field. As you might expect, the RootsOf() expression occurs often.

```
> p:=x^9+5*x^8+10*x^7+20*x^6+24*x^5-x^4+16;
```

$$p := x^9+5\,x^8+10\,x^7+20\,x^6+24\,x^5-x^4+16$$

```
> cp:=convert(1/p,fullparfrac,x);
```

$$cp := -\frac{1}{224}\frac{1}{x+2}+\left(\sum_{\_\alpha=\mathrm{RootOf}(\_Z^2+4)}\frac{\frac{39}{101440}\_\alpha-\frac{1}{6340}}{x-\_\alpha}\right)+\left(\sum_{\_\alpha=\mathrm{RootOf}(\_Z^6+3\,\_Z^5-\_Z+2)}\left(\frac{3186073}{3758098400}+\frac{1617079}{3758098400}\_\alpha^5+\frac{1671539}{3758098400}\_\alpha^4-\frac{1254161}{93952460}\_\alpha-\frac{280317}{1879049200}\_\alpha^3+\frac{2296667}{939524600}\_\alpha^2\right)/(x-\_\alpha)\right)$$

We can evaluate the first sum here, but there is not much we can do with the second one.

```
> convert(op(cp)[2],radical);
```

$$-\frac{\frac{1}{6340}-\frac{39}{50720}I}{x-2I}-\frac{\frac{1}{6340}+\frac{39}{50720}I}{x+2I}$$

In the case of denominators that are cubic, Maple can actually find the partial fraction expansion in radical form (though you may not like the answer):

```
> p2:=x^3-4*x^2+7*x+3;
```

$$p2 := x^3 - 4\,x^2 + 7\,x + 3$$

```
> cp2:=convert((x^2-2*x+4)/p2,fullparfrac,x);
```

$$cp2 := \sum_{\_\alpha = \mathrm{RootOf}(\_Z^3 - 4\,\_Z^2 + 7\,\_Z + 3)} \frac{-\frac{4}{315}\,\_\alpha^2 + \frac{47}{105} - \frac{5}{63}\,\_\alpha}{x - \_\alpha}$$

The RootsOf() expression is brief and very clear, but let's request an expression in terms of the actual roots:

```
> convert(cp2,radical);
```

$$\Bigg(-\frac{4}{315}\left(-\frac{1}{6}(820 + 180\sqrt{21})^{(1/3)} + \frac{\frac{10}{3}}{(820 + 180\sqrt{21})^{(1/3)}} + \frac{4}{3} \right)^2 + \frac{323}{945}$$

$$+ \frac{5}{378}(820 + 180\sqrt{21})^{(1/3)} - \frac{50}{189}\,\frac{1}{(820 + 180\sqrt{21})^{(1/3)}} \Bigg) \Bigg/ \Bigg($$

$$x + \frac{1}{6}(820 + 180\sqrt{21})^{(1/3)} - \frac{10}{3}\,\frac{1}{(820 + 180\sqrt{21})^{(1/3)}} - \frac{4}{3} \Bigg) + \Bigg(-\frac{4}{315}\Bigg($$

$$\frac{1}{12}(820 + 180\sqrt{21})^{(1/3)} - \frac{5}{3}\,\frac{1}{(820 + 180\sqrt{21})^{(1/3)}} + \frac{4}{3}$$

$$+ \frac{1}{2}\,I\sqrt{3}\left(-\frac{1}{6}(820 + 180\sqrt{21})^{(1/3)} - \frac{10}{3}\,\frac{1}{(820 + 180\sqrt{21})^{(1/3)}} \right)\Bigg)^2 + \frac{323}{945}$$

$$- \frac{5}{756}(820 + 180\sqrt{21})^{(1/3)} + \frac{\frac{25}{189}}{(820 + 180\sqrt{21})^{(1/3)}}$$

$$- \frac{5}{126}\,I\sqrt{3}\left(-\frac{1}{6}(820 + 180\sqrt{21})^{(1/3)} - \frac{10}{3}\,\frac{1}{(820 + 180\sqrt{21})^{(1/3)}} \right)\Bigg) \Bigg/ \Bigg(x$$

$$-\frac{1}{12}(820+180\sqrt{21})^{(1/3)}+\frac{\frac{5}{3}}{(820+180\sqrt{21})^{(1/3)}}-\frac{4}{3}$$

$$-\frac{1}{2}I\sqrt{3}\Bigg(-\frac{1}{6}(820+180\sqrt{21})^{(1/3)}-\frac{10}{3}\frac{1}{(820+180\sqrt{21})^{(1/3)}}\Bigg)\Bigg)\Bigg)+\Bigg(-\frac{4}{315}\Bigg($$

$$\frac{1}{12}(820+180\sqrt{21})^{(1/3)}-\frac{5}{3}\frac{1}{(820+180\sqrt{21})^{(1/3)}}+\frac{4}{3}$$

$$-\frac{1}{2}I\sqrt{3}\Bigg(-\frac{1}{6}(820+180\sqrt{21})^{(1/3)}-\frac{10}{3}\frac{1}{(820+180\sqrt{21})^{(1/3)}}\Bigg)\Bigg)^2+\frac{323}{945}$$

$$-\frac{5}{756}(820+180\sqrt{21})^{(1/3)}+\frac{\frac{25}{189}}{(820+180\sqrt{21})^{(1/3)}}$$

$$+\frac{5}{126}I\sqrt{3}\Bigg(-\frac{1}{6}(820+180\sqrt{21})^{(1/3)}-\frac{10}{3}\frac{1}{(820+180\sqrt{21})^{(1/3)}}\Bigg)\Bigg)\Bigg/\Bigg(x$$

$$-\frac{1}{12}(820+180\sqrt{21})^{(1/3)}+\frac{\frac{5}{3}}{(820+180\sqrt{21})^{(1/3)}}-\frac{4}{3}$$

$$+\frac{1}{2}I\sqrt{3}\Bigg(-\frac{1}{6}(820+180\sqrt{21})^{(1/3)}-\frac{10}{3}\frac{1}{(820+180\sqrt{21})^{(1/3)}}\Bigg)\Bigg)$$

```
>
```

Trigonometric Expressions

```
> restart;
```

The procedures convert() and combine() may be used to manipulate trigonometric expressions (and much else - check help).

```
> ex1:=sin(2*x): ex1=convert(ex1,tan);
```

$$\sin(2\,x)=2\,\frac{\tan(x)}{1+\tan(x)^2}$$

```
> ex1=convert(ex1,exp);
```

$$\sin(2\,x) = \frac{-1}{2}\,I\left(e^{(2\,I\,x)} - \frac{1}{e^{(2\,I\,x)}}\right)$$

```
> ex2:=tan(x)*sec(2*x)+sec(x): ex2=convert(ex2,sincos);
```

$$\tan(x)\sec(2\,x) + \sec(x) = \frac{\sin(x)}{\cos(x)\cos(2\,x)} + \frac{1}{\cos(x)}$$

```
> ex2=simplify(convert(ex2,tan));
```

$$\tan(x)\sec(2\,x) + \sec(x) = \frac{2\cos(x)^2 - 1 + \sin(x)}{\cos(x)\,(2\cos(x)^2 - 1)}$$

```
> ex2=simplify(convert(ex2,exp));
```

$$\tan(x)\sec(2\,x) + \sec(x) = 2\,\frac{e^{(I\,x)}\,(-I\,e^{(3\,I\,x)} + I\,e^{(I\,x)} + e^{(4\,I\,x)} + 1)}{(e^{(2\,I\,x)} + 1)\,(e^{(4\,I\,x)} + 1)}$$

The combine() procedure combines terms as you might expect, but it also may be used to obtain Fourier sum expansions (Fourier series if you like) of trigonometric functions.

```
> ex3:=Sum(cos(x)^k,k=1..4): ex3=combine(value(ex3));
```

$$\sum_{k=1}^{4} \cos(x)^k = \frac{7}{4}\cos(x) + \cos(2\,x) + \frac{7}{8} + \frac{1}{4}\cos(3\,x) + \frac{1}{8}\cos(4\,x)$$

```
> ex4:=Sum(sin(x)^k,k=1..4): ex4=combine(value(ex4));
```

$$\sum_{k=1}^{4} \sin(x)^k = \frac{7}{4}\sin(x) + \frac{7}{8} - \cos(2\,x) - \frac{1}{4}\sin(3\,x) + \frac{1}{8}\cos(4\,x)$$

```
> ex5:=cos(3*x)*sin(2*x)^2: ex5=combine(ex5);
```

$$\cos(3\,x)\sin(2\,x)^2 = \frac{1}{2}\cos(3\,x) - \frac{1}{4}\cos(x) - \frac{1}{4}\cos(7\,x)$$

Maple has a builtin database of trigonometric identies. We can consult the database by means of the trigsubs() procedure. (We can do much more - check help.)

```
> trigsubs(sin(2*x)*cos(x));
```

$$\left[\frac{1}{2}\sin(x) + \frac{1}{2}\sin(3\,x)\right]$$

```
> trigsubs(cos(x));
```

$$\left[\cos(x), \cos(-x), 2\cos\left(\frac{1}{2}x\right)^2 - 1, 1 - 2\sin\left(\frac{1}{2}x\right)^2, \cos\left(\frac{1}{2}x\right)^2 - \sin\left(\frac{1}{2}x\right)^2, \frac{1}{\sec(x)}, \frac{1}{\sec(-x)},\right.$$

$$\left[\frac{1-\tan\left(\frac{1}{2}x\right)^2}{1+\tan\left(\frac{1}{2}x\right)^2}, \frac{1}{2}\mathrm{e}^{(I\,x)}+\frac{1}{2}\mathrm{e}^{(-I\,x)}\right]$$

Comments on combine()

We can use combine() to combine terms of course. Consider

```
> combine(sqrt(x)*sqrt(y));
```

$$\sqrt{x}\,\sqrt{y}$$

Here Maple refuses the obvious simplification because it is not always correct. However, we can force it by passing the symbolic instruction to combine().

```
> combine(sqrt(x)*sqrt(y),symbolic);
```

$$\sqrt{x\,y}$$

If you make use of such "simplifications" then you will have to be careful.

```
>
```

Least Squares Fit (stats)

```
> restart;
```

The stats package provides a very convenient way to do least squares fitting in Maple. Actually we only need the subpackage fit. The subpackage statplots is also handy, though sometimes the package plots is more convenient.

We can also use the leastsqrs() procedure in the linalg package to do leastsquares fitting, but we would have to a bit more work since we first have to express our problem as a system of linear equations. The stats package handles all this for us.

```
> with(stats): with(plots):
Warning, the name changecoords has been redefined
```

In statistical applications there is often much data. You certainly do not want to type it all into your worksheet. Check help for the procedure importdata() to see one way to get data from a file. We will not need this procedure for our small examples here.

Here’s some data:

```
> Xdata:=[1.0,2.0,3.0,4.0,5.0,6.0,7.0,8.0,9.0];
```

$$Xdata := [1.0, 2.0, 3.0, 4.0, 5.0, 6.0, 7.0, 8.0, 9.0]$$

```
> Ydata:=[2.33,0.0626,-2.16,-2.45,-.357,2.21,2.75,0.636,-2.45];
```

$$Ydata := [2.33, .0626, -2.16, -2.45, -.357, 2.21, 2.75, .636, -2.45]$$

We need a list of data point for the plot() procedure. We can obtain such a list easily:

```
> data:=[seq([Xdata[k],Ydata[k]],k=1..nops(Xdata))];
```

$$data := [[1.0, 2.33], [2.0, .0626], [3.0, -2.16], [4.0, -2.45], [5.0, -.357], [6.0, 2.21], [7.0, 2.75], [8.0, .636], [9.0, -2.45]]$$

Let's plot the data points and save the plot for later use:

```
> dataplot:=plot(data,style=point,symbol=circle,symbolsize=16,col
  or=blue):
```

Let's assume we have some good reason to suppose that our data points should satisy the following trigonometric relation:

```
> eqn:=y=a+b*cos(x)+c*sin(x)+d*cos(2*x)+e*sin(2*x);
```

$$eqn := y = a + b\cos(x) + c\sin(x) + d\cos(2\,x) + e\sin(2\,x)$$

The five coefficients are to be determined. They are our variables.

```
> vars:={a,b,c,d,e};
```

$$vars := \{e, a, c, d, b\}$$

We are ready to ask Maple for the coefficients. Note I used floating point data in Xdata above, even though the values are integers. This is important. If I had used Xdata:={1,2,..] instead then Maple would do symbolic calculations involving expressions sin(1), cos(1), sin(2), and so on, for at least part of the calculation. The result would be that fiteqn would take much longer to calculate. Try it.

```
> fiteqn:=fit[leastsquare[[x,y],eqn,vars]]([Xdata,Ydata]);
```

$$fiteqn := y = .00008970255944 + 2.617965973\cos(x) + 1.074450851\sin(x) - .01316995873\cos(2\,x) + .07966783881\sin(2\,x)$$

Let's plot the calculated least squares fit and then display it with the original data:

```
> fitplot:=plot(rhs(fiteqn),x=0..10,color=red,thickness=2):
> display({dataplot,fitplot});
```



It is a nice fit! But do note, all this would be meaningless unless we have some good reason to expect the functional relation eqn is correct for our data.

>

Least Squares Fit (linalg)

```
> restart;
```

Let's consider the same problem as in the previous section, but this time we will use the linalg package.

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and
unprotected
> Xdata:=[1.0,2.0,3.0,4.0,5.0,6.0,7.0,8.0,9.0];
```

$$Xdata := [1.0, 2.0, 3.0, 4.0, 5.0, 6.0, 7.0, 8.0, 9.0]$$

```
> Ydata:=[2.33,0.0626,-2.16,-2.45,-.357,2.21,2.75,0.636,-2.45];
```

$$Ydata := [2.33, .0626, -2.16, -2.45, -.357, 2.21, 2.75, .636, -2.45]$$

Here's our model, as before:

```
> eqn:=y=a+b*cos(x)+c*sin(x)+d*cos(2*x)+e*sin(2*x);
```

$$eqn := y = a + b\cos(x) + c\sin(x) + d\cos(2x) + e\sin(2x)$$

Now we find the linear equations that the coefficients should satisfy (approximately) by substitution:

```
> for k from 1 to nops(Xdata) do
  lineqn[k]:=subs(x=Xdata[k],y=Ydata[k],eqn); od:
```

Then we put the equations in a list:

```
> eqns:=[seq(lineqn[k],k=1..nops(Xdata))];
```

$$eqns := [2.33 = a + .5403023059\, b + .8414709848\, c - .4161468365\, d + .9092974268\, e,$$
$$.0626 = a - .4161468365\, b + .9092974268\, c - .6536436209\, d - .7568024953\, e,$$
$$-2.16 = a - .9899924966\, b + .1411200081\, c + .9601702867\, d - .2794154982\, e,$$
$$-2.45 = a - .6536436209\, b - .7568024953\, c - .1455000338\, d + .9893582466\, e,$$
$$-.357 = a + .2836621855\, b - .9589242747\, c - .8390715291\, d - .5440211109\, e,$$
$$2.21 = a + .9601702867\, b - .2794154982\, c + .8438539587\, d - .5365729180\, e,$$
$$2.75 = a + .7539022543\, b + .6569865987\, c + .1367372182\, d + .9906073557\, e,$$
$$.636 = a - .1455000338\, b + .9893582466\, c - .9576594803\, d - .2879033167\, e,$$
$$-2.45 = a - .9111302619\, b + .4121184852\, c + .6603167082\, d - .7509872468\, e]$$

Once we have a list of linear equations we can ask Maple to compute the coefficient matrix by means of the procedure genmatrix(). Note that we have to specify the unknowns in a list. It is easy enough to extract the inhomogeneous term by hand, but the genmatrix() procedure will do that as well if we provide an optional third parameter to assign it to.

```
> A:=genmatrix(eqns,[a,b,c,d,e],b);
```

$$A := \begin{bmatrix} -1 & -.5403023059 & -.8414709848 & .4161468365 & -.9092974268 \\ -1 & .4161468365 & -.9092974268 & .6536436209 & .7568024953 \\ -1 & .9899924966 & -.1411200081 & -.9601702867 & .2794154982 \\ -1 & .6536436209 & .7568024953 & .1455000338 & -.9893582466 \\ -1 & -.2836621855 & .9589242747 & .8390715291 & .5440211109 \\ -1 & -.9601702867 & .2794154982 & -.8438539587 & .5365729180 \\ -1 & -.7539022543 & -.6569865987 & -.1367372182 & -.9906073557 \\ -1 & .1455000338 & -.9893582466 & .9576594803 & .2879033167 \\ -1 & .9111302619 & -.4121184852 & -.6603167082 & .7509872468 \end{bmatrix}$$

```
> evalm(b);
```

$$[-2.33, -.0626, 2.16, 2.45, .357, -2.21, -2.75, -.636, 2.45]$$

As you might expect the system of equations above is inconsistent:

```
> linsolve(A,b);
```

Maple indicates the lack of a solution by returning an empty response. For the approximate solution that is best in the sense of least squares we use the leastsqrs() procedure.

```
> leastsqrs(A,ihterm);
Error, (in leastsqrs) improper arguments
```

These coefficients are the same as the ones obtained in the previous section.

```
>
```

Limits

```
> restart;
```

Maple is capable of computing a few limits. Note the inert form Limit() of the limit() procedure.

```
> limit(sin(x)/tan(x),x=0);
```

$$1$$

We can play the usual game to format the result nicely:

```
> Limit((cos(x)-1+x^2/2)*x^(-4),x=0): %=value(%);
```

$$\lim_{x \to 0} \frac{\cos(x) - 1 + \frac{1}{2}x^2}{x^4} = \frac{1}{24}$$

Here are some more example:

```
> limit(1/x^2,x=infinity);
```

$$0$$

```
> limit(x^x,x=0);
```

$$1$$

```
> limit(sin(1/x),x=0);
```

$$-1 .. 1$$

That is a curious result. Instead of saying that the limit does not exist Maple returns a range consisting of all of the limit points.

```
> limit(x*sin(1/x),x=0);
```

$$0$$

```
> limit((1+sin(x))^(1/x),x=0);
```

$$e$$

```
> limit(cos(x)^(1/x^2),x=0);
```

$$e^{(-1/2)}$$

Maple can also handle one-sided limits.

```
> Limit(sin(x)^(1/sin(x)^5),x=0): %=value(%);
```

$$\lim_{x \to 0} \sin(x)^{\left(\frac{1}{\sin(x)^5}\right)} = \mathit{undefined}$$

```
> Limit(sin(x)^(1/sin(x)^5),x=0,right): %=value(%);
```

$$\lim_{x \to 0+} \sin(x)^{\left(\frac{1}{\sin(x)^5}\right)} = 0$$

```
> Limit(sin(x)^(1/sin(x)^5),x=0,left): %=value(%);
```

$$\lim_{x \to 0-} \sin(x)^{\left(\frac{1}{\sin(x)^5}\right)} = \infty$$

Maple knows about some pretty sophisticated limits:

```
> Limit(Sum(1/k,k=1..n)-log(n),n=infinity): %=value(%);
```

$$\lim_{n \to \infty} \left(\sum_{k=1}^{n} \frac{1}{k}\right) - \ln(n) = \gamma$$

```
> 'gamma'=evalf(gamma,60);
```

$$\gamma = .577215664901532860606512090082402431042159335939923598805767$$

Problems

```
> restart;
```

You may have to load some libraries to do some of the problems. That's part of the problems!

Problem 1

Expand in partial fractions

```
> expr1:=(2*x-3)/((x+b)*(x+1));
```

$$\mathit{expr1} := \frac{2\,x-3}{(x+b)\,(x+1)}$$

What happens when b = 1? Find the limit as b goes to 1 for your expansion.

Problem 2

Find the Fourier series expansion of

```
> expr2:=cos(x)*cos(3*x)*cos(5*x)*cos(7*x)*cos(9*x);
```

$$expr2 := \cos(x)\cos(3x)\cos(5x)\cos(7x)\cos(9x)$$

Problem 3

Here's the human population p(t) of the Earth for a few years:

```
> DataPoints := [
[1900,1650],[1910,1750],[1920,1860],[1930,2070],[1940,2300],[19
50,2555],[1951,2593],[1952,2635],[1953,2680],[1954,2728],[1955,
2780],[1956,2833],[1957,2888],[1958,2945],[1959,2997],[1960,303
9],[1961,3080],[1962,3136],[1963,3206],[1964,3277],[1965,3346],
[1966,3416],[1967,3486],[1968,3558],[1969,3632],[1970,3708],[19
71,3785],[1972,3862],[1973,3939],[1974,4014],[1975,4088],[1976,
4160],[1977,4233],[1978,4305],[1979,4381],[1980,4457],[1981,453
3],[1982,4613],[1983,4694],[1984,4774],[1985,4855],[1986,4938],
[1987,5023],[1988,5110],[1989,5196],[1990,5284],[1991,5367],[19
92,5450],[1993,5531],[1994,5611],[1995,5691],[1996,5769],[1997,
5847],[1998,5925],[1999,6003],[2000,6080],[2001,6157],[2002,623
4] ] :
```

For the logistic model of population growth we have

```
> q=a*p-b*p^2;
```

$$q = a\,p - b\,p^2$$

where

```
> q(t)=diff(p(t),t);
```

$$q(t) = \frac{\partial}{\partial t}\,p(t)$$

Use the data above to estimate q and then do a least squares fit of $q = a\,p - b\,p^2$ to estimate a and b. Finally estimate the limiting population of the Earth (a/b in the logistic model).

```
>
>
>
```